


Lezione 7

In questa nuova puntata, impareremo a far generare al PIC dei segnali di bassa frequenza, prima semplici, poi più complessi. Precisiamo subito che, a meno di non disporre di un convertitore digitale/analogico, le forme d'onda prelevabili su un pin del controller non potranno che essere di tipo quadrato, poi facilmente modellabili con appositi filtri bassa-basso.

Passiamo quindi a vedere in figura 1 lo schema elettrico necessario alla prova dei software che andremo a proporre successivamente: al chip è stato collegato un pulsante (P1) ed un altoparlante, utilizzando così due sole linee del controller. L'alimentazione è sempre ricavata da un regolatore a tre terminali tipo 7805.

L'oscillatore è sempre di tipo quarzato, ma poiché non è indispensabile un'ottima precisione, potrete anche sostituirlo con uno meno costoso di tipo RC.

La soluzione adottata per il collegamento dell'altoparlante è senza dubbio la più semplice possibile, ma non offre l'elevata potenza necessaria per alimentare una tromba esponenziale oppure un piezo. Per far ciò dovrete sostituire il transistor, che è di tipo BC337, con un amplificatore più potente. Abbiamo volutamente applicato questa soluzione perché in poco tempo e con pochi componenti dà subito un'idea del suono generato dal PIC.



Il singolo tono

Passiamo allora subito a vedere il programma che permette la generazione di un singolo tono, ovvero una frequenza fissa:

```
TITLE 'PROG11: Tono a 1KHz'
list F=INHX8M,P=16C84
```

```
-----
;
RTCC EQU 01H ; Real Time Clock Counter
PCL EQU 02H ; Program Counter
STAT EQU 03H ; Registro di stato
PORTA EQU 05H ; Porta A
PORTB EQU 06H ; Porta B
EDATA EQU 08H ; Data EEPROM
EADR EQU 09H ; Address EEPROM
ECON1 EQU 88H ; Stato delle operaz. in EEPROM
ECON2 EQU 89H ; Vedi ECON1
INTCON EQU 0BH ; Registro abilitazione interrupt
```

```

TR_A EQU 85H ; Tris A
TR_B EQU 86H ; Tris B
OPTIO EQU 81H ; Registro OPTION
;-----
FR01 EQU 0CH ;
FR02 EQU 0DH ;
FR03 EQU 0EH ;
;-----
#define P1 PORTB,7 ; Pulsante 1
#define TONO PORTA,2 ; Uscita segnale BF
;-----
ORG 0
goto START ;Reset vector
nop
nop
goto INTERP ;Interrupt vector
;-----
; START PROGRAM
;-----
START bsf STAT,5 ; Seleziona SRAM banco 1
movlw b'0000' ; RA out
movwf TR_A ;
movlw b'11110000' ; RB0..RB3 out RB4..RB7 in
movwf TR_B ;
clrf INTCON ; Disabilita interrupt
movlw b'01000110' ; Enable pull-up su porta B
movwf OPTIO ; Copia W in OPTION
bcf STAT,5 ; Seleziona SRAM banco 0
movlw .128 ; Settaggio iniziale RTCC
movwf RTCC ;
clrf PORTA ; Poni a zero le uscite su A
clrf PORTB ; Poni a zero le uscite su B
;----- main program -----
MAIN clrw dt ; Azzerò watchdog
bcf TONO ; Tono off
btfsc P1 ; Testo se P1 premuto
goto MAIN ; Pulsante non premuto
;
LPTONO movlw b'0100' ; Togglo tono
xorwf PORTA,1 ; su porta A
movlw .100 ; Carico 100 in W
movwf FR01 ; Carico W in FR01
LP1 clrw dt ; Azzerò watchdog
nop ; Nessuna operazione
decfsz FR01 ; Decrem. FR01 e testo se zero
goto LP1 ; Non zero
btfsc P1 ; Testo se P1 ancora
goto MAIN ; Pulsante non premuto
goto LPTONO ;
;-----
INTERP ; Vettore interruzioni
retfie ;
;-----
END

```

Andiamo a vedere come viene implementata la frequenza di circa 1 KHz prefissata, tralasciando sia le impostazioni che la configurazione iniziale, già ampiamente illustrate nei mesi scorsi e passiamo subito alla parte di programma che genera il tono. Alla label MAIN si azzerava il watchdog, poi si pone a zero l'uscita relativa all'altoparlante e si testa il pulsante P1. Fino a quando questo pulsante non viene premuto, il PIC rimane a "loopare" su queste istruzioni appena descritte.

Non appena P1 viene premuto, il controller entra nel loop di generazione tono, inizializzato con la label LPTONO. Le prime due istruzioni, consentono, come abbiamo già visto in altre occasioni, di invertire lo stato dell'uscita relativa all'altoparlante, nel nostro caso la RA2. Così, se prima avevamo uno zero, adesso ci troviamo un uno.

Poi viene caricato il registro FR01 con il valore decimale 100. A questo punto è stato introdotto un loop di ritardo che fa capo alla label LP1. Il ritardo introdotto è, come sappiamo, di 5 (numero cicli) x 1 (tempo per un ciclo in microsecondi) x 100 (numero impostato nel registro FR01), ovvero in totale 500 microsecondi. Al termine di questo ritardo, viene nuovamente testato il pulsante P1 per vedere se è sempre premuto. In caso affermativo, si torna alla label LPTONO, altrimenti si torna al ciclo iniziale.

Facendo due conti, si vede che la frequenza di uscita sul pin RA2 del PIC sarà di circa $1/(500 + 500)$ ovvero di 1.000 Hz.

Provate a variare il valore caricato nel registro FR01 e vedrete che anche la frequenza di uscita varierà. Quale saranno le due frequenze minima e massima che potrete ottenere da questo programma? A voi la risposta.

-continua.